WST 221: Practical — Examples: PROC IML

- PROC IML(Interactive Matrix Language) is a procedure in SAS that enables us to work with matrices.
- You can also look in the HELP menu of SAS under CONTENTS:

  SAS Products <= SAS/IML User's Guide <= Language Reference

- Consider the following two examples:

1. **Define the following matrices:**

$$A = \begin{pmatrix} 1 & 5 \\ 3 & 8 \end{pmatrix}, \quad B = \begin{pmatrix} 5 & 0 \\ 2 & -1 \end{pmatrix}, \quad C = \begin{pmatrix} 1 & 2 \\ -3 & 4 \\ 2 & 6 \end{pmatrix} \quad \text{and} \quad D = \begin{pmatrix} 4 & 9 & 16 \end{pmatrix}$$

```
proc iml;
print 'Define the matrices';
A={1 5,
 3 8};
B={5 0,
 2 -1};
C={1 2,
-3 4,
 2 6};
D={4 9 16};
print A B C D;
```

```
Define the matrices

   A                    B                    C
   1         5          5          0         1         2
   3         8          2         -1        -3         4
                                              2         6

         D
   4          9         16
```

   (a) **Operators:**

```
print 'Operators';
E=A+B;
F=A-B;
G=A*B;
H=3*B;
K=C';
L=A/10;
M=7**2;
N=C[2,1];
P=A#B;
Q=B/A;
R=A##B;
S=C||D';
T=K//D;
print E F G H K L M N P Q R S T;
```

```
Operators

   E                 F                 G
   6         5      -4         5      15        -5
   5         7       1         9      31        -8

   H                 K                           L
  15         0       1        -3       2       0.1       0.5
   6        -3       2         4       6       0.3       0.8

   M         N       P                 Q
  49        -3       5         0       5         0
                     6        -8 0.6666667    -0.125

   R                 S
   1         1       1         2       4
   9     0.125      -3         4       9
                     2         6      16

         T
   1        -3       2
   2         4       6
   4         9      16
```

(b) **Statements, Functions and Subroutines**

```
print 'Statements, Functions and Subroutines';
sqrtD=sqrt(D);
invA=inv(A);
detA=det(A);
I2=I(2);
diagD=diag(D);
vecdiagA=vecdiag(A);
zero=J(2,3,0);
one=J(3,2,1);
two=J(3,1,2);
obs=cusum(J(3,1,1));
print sqrtD invA detA,,I2 diagD vecdiagA,,zero one two,,obs;
```

```
Statements, Functions and Subroutines

sqrtD                          invA                detA
   2         3         4 -1.142857 0.7142857        -7
                         0.4285714 -0.142857

  I2                diagD                          vecdiagA
   1         0         4         0         0         1
   0         1         0         9         0         8
                       0         0        16

zero                             one                 two
   0         0         0         1         1         2
   0         0         0         1         1         2
                                 1         1         2

 obs
      1
      2
      3
```

(c) **Base SAS Functions Accessible from SAS/IML**

i. **Mathematical Functions**

```
print 'Base SAS Functions Accessible from  SAS/IML';
print 'Mathematical Functions';
pi=(gamma(0.5))**2;
ln5=log(5);
log5=log10(5);
exp5=exp(5);
print pi ln5 log5 exp5;
```

```
Base SAS Functions Accessible from  SAS/IML
Mathematical Functions

  pi      ln5      log5      exp5
3.1415927 1.6094379   0.69897 148.41316
```

ii. **Probability Functions**

```
print 'Probability Functions';
p1=probnorm(1.96);          p2=probchi(3.94,10);
p3=probf(3.62,3,5);         p4=probt(2.262,9);
print p1 [format=5.3] p2 [format=5.3] p3 [format=5.3] p4 [format=5.3];
```

```
Probability Functions

p1    p2    p3    p4
0.975 0.050 0.900 0.975
```

iii. **Quantile Functions**

```
print 'Quantile Functions';
q1=probit(0.975);   q2=cinv(0.95,9);
q3=finv(0.9,5,4);   q4=tinv(0.99,14);
print q1 [format=6.3] q2 [format=6.3] q3 [format=6.3] q4 [format=6.3];
```

```
Quantile Functions

q1     q2     q3     q4
1.960 16.919  4.051  2.624
```

iv. **Random Number Functions**

```
print 'Random Number Functions';
mu=10; sigma=2; theta=5; kappa=10; nu=8; n=10; p=0.5;
r1=ranuni(111);
r2=rannor(222);
r3=theta*rangam(333,kappa);
r4=2*rangam(444,nu/2);
r5=J(4,1,mu)+sigma*rannor(J(4,1,555));
r6=2*rangam(J(3,1,666),J(3,1,nu/2));
r7=ranbin(J(4,1,777),J(4,1,n),J(4,1,p));
print r1 r2 r3 r4 r5 r6 r7;
```

```
Random Number Functions

     r1        r2        r3        r4        r5        r6        r7
0.5308453 -0.642282 68.067601 9.1767481 4.4045789 9.0687763         5
                                        12.582506 4.0576524         6
                                        9.4329906 13.307998         7
                                        10.318969                   7
```

2. A total of $r = 100$ samples of size $n = 4$ are generated from a $N(60, 10^2)$ population. The sample mean for each of the $r = 100$ samples is calculated. The empirical distribution of the sample mean $\bar{X}$ is investigated by making use of PROC UNIVARIATE.

SAS Program:

```
proc iml;
r=100; n=4; mu=60; sigma=10;

matrix=J(r,4,0);
do i=1 to r;
x=J(n,1,mu)+sigma*rannor(J(n,1,123));
xbar=J(1,n,1)*x/n;
min=min(x); max=max(x); sum=sum(x);
matrix[i,1]=xbar; matrix[i,2]=min; matrix[i,3]=max; matrix[i,4]=sum;
print i x xbar min max sum;
end;
print matrix;
create d from matrix[colname={'xbar' 'min' 'max' 'sum'}];
append from matrix;

proc univariate data=d plot normal;
var xbar;
histogram / normal(color=(red blue) mu=60 est sigma=5 est);
cdfplot / normal(mu=100 sigma=5 color=red);
run;
```

SAS Output: (From PROC UNIVARIATE)

```
Parameters for Normal Distribution
Parameter   Symbol   Estimate
Mean        Mu             60
Std Dev     Sigma           5

Quantiles for Normal Distribution
     ------Quantile------
Percent   Observed   Estimated

1.0    47.6072     48.3683
5.0    50.5404     51.7757
10.0    53.3194     53.5922
25.0    55.9856     56.6276
50.0    59.3019     60.0000
75.0    62.2268     63.3724
90.0    64.3097     66.4078
95.0    66.0882     68.2243
99.0    69.5406     71.6317

Parameters for Normal Distribution
Parameter   Symbol   Estimate
Mean        Mu       59.09674
Std Dev     Sigma    4.563275

Quantiles for Normal Distribution
     ------Quantile------
Percent   Observed   Estimated

1.0    47.6072     48.4810
5.0    50.5404     51.5908
10.0    53.3194     53.2487
25.0    55.9856     56.0189
50.0    59.3019     59.0967
75.0    62.2268     62.1746
90.0    64.3097     64.9448
95.0    66.0882     66.6027
99.0    69.5406     69.7125
```